In[1]:=
```
NotebookDirectory[]
```

Out[1]=
```
C:\Dropbox\Work\myweb\Courses\Math_pages\Math_312\
```

A bijection from ℕ onto ℕxℕ.

The next formula gives an explicit form of the "repeat sequence".

In[2]:=
```
Clear[r, n];
r[n_] := Floor[Sqrt[2 n] + 1/2]
```

In[4]:=
```
Table[r[n], {n, 1, 50}]
```

Out[4]=
```
{1, 2, 2, 3, 3, 3, 4, 4, 4, 4, 5, 5, 5, 5, 5, 6,
 6, 6, 6, 6, 6, 7, 7, 7, 7, 7, 7, 7, 8, 8, 8, 8, 8, 8,
 8, 8, 9, 9, 9, 9, 9, 9, 9, 9, 9, 10, 10, 10, 10, 10}
```

Recursive formula for the above repeat sequence is

In[5]:=
```
Clear[rr];
rr[1] = 1; rr[n_] := rr[n] = 1 + rr[n - rr[n - 1]]
```

In[6]:=
```
rr[2]
```

Out[6]=
```
2
```

In[7]:=
```
Table[rr[k], {k, 1, 120}]
```

Out[7]=
```
{1, 2, 2, 3, 3, 3, 4, 4, 4, 4, 5, 5, 5, 5, 5, 6, 6, 6, 6, 6, 6, 7,
 7, 7, 7, 7, 7, 7, 8, 8, 8, 8, 8, 8, 8, 8, 9, 9, 9, 9, 9, 9, 9, 9,
 9, 10, 10, 10, 10, 10, 10, 10, 10, 10, 10, 11, 11, 11, 11, 11,
 11, 11, 11, 11, 11, 11, 12, 12, 12, 12, 12, 12, 12, 12, 12,
 12, 12, 12, 13, 13, 13, 13, 13, 13, 13, 13, 13, 13, 13, 13,
 13, 14, 14, 14, 14, 14, 14, 14, 14, 14, 14, 14, 14, 14, 14,
 15, 15, 15, 15, 15, 15, 15, 15, 15, 15, 15, 15, 15, 15, 15}
```

The following formula defines the sequence of triangular numbers.

In[8]:=
```
Clear[T, n];
T[n_] := 1/2 n (n + 1)
```

In[10]:=
```
Table[T[n], {n, 0, 50}]
```

Out[10]=
```
{0, 1, 3, 6, 10, 15, 21, 28, 36, 45, 55, 66, 78, 91, 105, 120,
 136, 153, 171, 190, 210, 231, 253, 276, 300, 325, 351, 378,
 406, 435, 465, 496, 528, 561, 595, 630, 666, 703, 741, 780,
 820, 861, 903, 946, 990, 1035, 1081, 1128, 1176, 1225, 1275}
```

Now, we can give an explicit formula for a **bijection** from ℕ onto ℕxℕ.

In[11]:=
```
Clear[n, Bf];
Bf[n_] := {(n - T[r[n] - 1]), (T[r[n]] - n + 1)}
```

In[13]:=
```
Bf[179]
```

Out[13]=
```
{8, 12}
```

This is the table of the first 100 values of **Bf** :

In[14]:=
```
Table[Bf[k], {k, 1, 100}]
```

Out[14]=
```
{{1, 1}, {1, 2}, {2, 1}, {1, 3}, {2, 2}, {3, 1}, {1, 4}, {2, 3},
 {3, 2}, {4, 1}, {1, 5}, {2, 4}, {3, 3}, {4, 2}, {5, 1}, {1, 6},
 {2, 5}, {3, 4}, {4, 3}, {5, 2}, {6, 1}, {1, 7}, {2, 6},
 {3, 5}, {4, 4}, {5, 3}, {6, 2}, {7, 1}, {1, 8}, {2, 7},
 {3, 6}, {4, 5}, {5, 4}, {6, 3}, {7, 2}, {8, 1}, {1, 9},
 {2, 8}, {3, 7}, {4, 6}, {5, 5}, {6, 4}, {7, 3}, {8, 2},
 {9, 1}, {1, 10}, {2, 9}, {3, 8}, {4, 7}, {5, 6}, {6, 5},
 {7, 4}, {8, 3}, {9, 2}, {10, 1}, {1, 11}, {2, 10}, {3, 9},
 {4, 8}, {5, 7}, {6, 6}, {7, 5}, {8, 4}, {9, 3}, {10, 2},
 {11, 1}, {1, 12}, {2, 11}, {3, 10}, {4, 9}, {5, 8}, {6, 7},
 {7, 6}, {8, 5}, {9, 4}, {10, 3}, {11, 2}, {12, 1}, {1, 13},
 {2, 12}, {3, 11}, {4, 10}, {5, 9}, {6, 8}, {7, 7}, {8, 6},
 {9, 5}, {10, 4}, {11, 3}, {12, 2}, {13, 1}, {1, 14}, {2, 13},
 {3, 12}, {4, 11}, {5, 10}, {6, 9}, {7, 8}, {8, 7}, {9, 6}}
```

The inverse of the function **Bf** is the following function

In[15]:=
```
Clear[Af, s, t];
(* Af[{s_,t_}]:=(s+t-2) (s+t-1)/2+s *)
Af[{s_, t_}] := (s + t - 1) (s + t)/2 - t + 1
```

In[17]:=
```
Af[{1, 1}]
```

Out[17]= 1

In[18]:=
```
Af[{1, 2}]
```

Out[18]= 2

In[19]:=
```
Af[{2, 1}]
```

Out[19]= 3

In[20]:=
```
Af[{8, 12}]
```

Out[20]= 179

To illustrate that the functions **Bf** and **Af** are inverses of each other calculate

First **Af[Bf[*n*]]** for many values

In[21]:=
```
Af[Bf[12]]
```

Out[21]= 12

In[22]:=
```
Bf[Af[{123, 54}]]
```

Out[22]= {123, 54}

Now do it for 100 values.

In[23]:=
```
Table[{Af[Bf[k]], k}, {k, 1, 100}]
```

Out[23]=
```
{{1, 1}, {2, 2}, {3, 3}, {4, 4}, {5, 5}, {6, 6}, {7, 7},
 {8, 8}, {9, 9}, {10, 10}, {11, 11}, {12, 12}, {13, 13},
 {14, 14}, {15, 15}, {16, 16}, {17, 17}, {18, 18}, {19, 19},
 {20, 20}, {21, 21}, {22, 22}, {23, 23}, {24, 24}, {25, 25},
 {26, 26}, {27, 27}, {28, 28}, {29, 29}, {30, 30}, {31, 31},
 {32, 32}, {33, 33}, {34, 34}, {35, 35}, {36, 36}, {37, 37},
 {38, 38}, {39, 39}, {40, 40}, {41, 41}, {42, 42}, {43, 43},
 {44, 44}, {45, 45}, {46, 46}, {47, 47}, {48, 48}, {49, 49},
 {50, 50}, {51, 51}, {52, 52}, {53, 53}, {54, 54}, {55, 55},
 {56, 56}, {57, 57}, {58, 58}, {59, 59}, {60, 60}, {61, 61},
 {62, 62}, {63, 63}, {64, 64}, {65, 65}, {66, 66}, {67, 67},
 {68, 68}, {69, 69}, {70, 70}, {71, 71}, {72, 72}, {73, 73},
 {74, 74}, {75, 75}, {76, 76}, {77, 77}, {78, 78}, {79, 79},
 {80, 80}, {81, 81}, {82, 82}, {83, 83}, {84, 84}, {85, 85},
 {86, 86}, {87, 87}, {88, 88}, {89, 89}, {90, 90},
 {91, 91}, {92, 92}, {93, 93}, {94, 94}, {95, 95},
 {96, 96}, {97, 97}, {98, 98}, {99, 99}, {100, 100}}
```

Now ask Mathematica to confirm that the composition is equal to the identity function for the first 100 values:

In[24]:=
```
Table[Af[Bf[k]] == k, {k, 1, 100}]
```

Out[24]=
```
{True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True}
```

Let Mathematica do checking

In[25]:=
```
Apply[And, Table[Af[Bf[k]] == k, {k, 1, 100}]]
```

Out[25]=
```
True
```

Pay attention, the table below is large:

In[26]:=
```
Apply[And, Table[Af[Bf[k]] == k, {k, 1, 100 000}]]
```

Out[26]=
```
True
```

Now **Bf[Af[{*s, t*}]]** for many values

In[27]:=
```
Bf[Af[{134, 216}]]
```

Out[27]=
```
{134, 216}
```

In[28]:=
```
Flatten[Table[Bf[Af[{j, k}]] == {j, k}, {k, 1, 10},
  {j, 1, 10}], 1]
```

Out[28]=
```
{True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True,
 True, True, True, True, True, True, True, True, True, True}
```

Or, many more values, with Mathematica doing the checking

In[29]:=
```
Apply[And,
 Flatten[Table[Bf[Af[{j, k}]] == {j, k}, {k, 1, 500},
   {j, 1, 500}], 1]]
```

Out[29]=
```
True
```

The following two functions will help illustrate how listing of points in ℕxℕ works in the coordinate system.

In[30]:=
```
Clear[Illustrate, Illustrate1, Illustrate2, s, t, m, n];

Illustrate[m_, n_] := Graphics[{
   {Thickness[0.004], GrayLevel[0.7],
    {Line[{{#, -1}, {#, 100}}],
       Line[{{-1, #}, {100, #}}]} & /@ Range[0, 30]},
   {PointSize[0.04], Point[Bf[#]] & /@ Range[1, n]},
   {Text[Af[Bf[#]], Bf[#],
       BaseStyle → {FontSize → 7, FontColor → White,
         FontWeight → "Bold"}]} & /@ Range[1, m]
  }]

Illustrate1[n_] := Graphics[{
   {Thickness[0.004], GrayLevel[0.7],
    {Line[{{#, -1}, {#, 100}}],
       Line[{{-1, #}, {100, #}}]} & /@ Range[0, 30]},
   {PointSize[0.04], Point[Bf[n]]},
   {Text[Af[Bf[n]], Bf[n],
     BaseStyle → {FontSize → 7, FontColor → White,
       FontWeight → "Bold"}]}
  }]

Illustrate2[{s_, t_}] := Graphics[{
   {Thickness[0.004], GrayLevel[0.7],
    {Line[{{#, -1}, {#, 100}}],
       Line[{{-1, #}, {100, #}}]} & /@ Range[0, 30]},
   {PointSize[0.04], Point[{s, t}]},
   {Text[Af[{s, t}], {s, t},
     BaseStyle → {FontSize → 7, FontColor → White,
       FontWeight → "Bold"}]}
  }]
```

In[34]:=
```
Show[
 Illustrate1[216],
 PlotRange → {{0.5, 20.5}, {0.5, 20.5}}, Frame → True,
 AspectRatio → Automatic, ImageSize → 350,
 FrameTicks → {Range[1, 20], Range[1, 20], {}, {}}
]
```

Out[34]=



In[35]:=
```
Show[
 Illustrate2[{6, 14}],
 PlotRange → {{0.5, 20.5}, {0.5, 20.5}}, Frame → True,
 AspectRatio → Automatic, ImageSize → 350,
 FrameTicks → {Range[1, 20], Range[1, 20], {}, {}}
]
```

Out[35]=



In[36]:=
```
Show[
 Illustrate1[177],
 PlotRange → {{0.5, 20}, {0.5, 20}}, Frame → True,
 AspectRatio → Automatic, ImageSize → 350,
 FrameTicks → {Range[1, 20], Range[1, 20], {}, {}}
]
```

Out[36]=



Now the animation of listing of the first two hundred points

In[37]:=
```
ListAnimate[Table[Show[
    Illustrate[n, n],
    PlotRange → {{0.5, 20.5}, {0.5, 20.5}}, Frame → True,
    AspectRatio → Automatic, ImageSize → 350,
    FrameTicks → {Range[1, 20], Range[1, 20], {}, {}}
  ], {n, 1, 200}]]
```

Out[37]=



In[38]:=
```
Show[
 Illustrate[685, 685],
 PlotRange → {{0.5, 20.5}, {0.5, 20.5}}, Frame → True,
 AspectRatio → Automatic, ImageSize → 350,
 FrameTicks → {Range[1, 20], Range[1, 20], {}, {}}
]
```

Out[38]=